Maple-assisted proof of formula for A207084

Robert Israel

15 May 2018

There are $2^8 = 256$ configurations for a 2×4 sub-array, but only 169 avoid [0,0,0] horizontally. Consider the 169×169 transition matrix T such that $T_{ij} = 1$ if the bottom two rows of a 3×4 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the middle row is compatible with both i and j, and no column is $[0, 0, 1]$), and 0 otherwise. The following Maple code computes it. Configurations are encoded as 8-element lists in the order

$$\begin{bmatrix} 1 & 2 & 3 & 4 \\ 5 & 6 & 7 & 8 \end{bmatrix}$$

```
> Configs:= select(t -> min(t[1]+t[2]+t[3],t[2]+t[3]+t[4],t[5]+t[6]
  +t[7],t[6]+t[7]+t[8])>=1, [seq(convert(2^8+i,base,2)[1..8],i=0.
  .2^8-1)]): nops(Configs);
```

$$169 \tag{1}$$

```
> Compatible:= proc(i,j) local k;
    if Configs[i][1..4] <> Configs[j][5..8] then return 0 fi;
    for k from 0 to 3 do if [Configs[j][1+k],Configs[j][5+k],
  Configs[i][5+k]]=[0,0,1] then return 0 fi od;
    1
  end proc:
> T:= Matrix(169,169,Compatible):
```

Thus for $n \geq 2$, $a(n) = u\, T^{n-2}\, v$ where u and v are row and column vectors respectively of all 1's. The following Maple code produces these vectors.

```
> u:= Vector[row](169, 1):
  v:= Vector(169, 1):
```

To check, here are the first few entries of our sequence.

```
> TV[0]:= v:
  for n from 1 to 14 do TV[n]:= T . TV[n-1] od:
> A:= [13, seq(u . TV[n],n=0..14)];
```

$$A := [13, 169, 1393, 10621, 75221, 518001, 3500117, 23428181, 155913829, 1034324253, 6848794157, 45301138173, 299456026377, 1978795266229, 13073066599357, 86357724891721] \tag{2}$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= LinearAlgebra:-MinimalPolynomial(T, t);
```

$$P := t^{28} - 5\,t^{27} - 46\,t^{26} + 156\,t^{25} + 956\,t^{24} - 1666\,t^{23} - 10487\,t^{22} + 5837\,t^{21} + 60933\,t^{20} + 8571\,t^{19} - 188632\,t^{18} - 98162\,t^{17} + 321573\,t^{16} + 236797\,t^{15} - 312253\,t^{14} - 273827\,t^{13} + 173536\,t^{12} + 172166\,t^{11} - 53745\,t^{10} - 60465\,t^{9} + 8887\,t^{8} + 11757\,t^{7} - 756\,t^{6} - 1218\,t^{5} + 34\,t^{4} + 60\,t^{3} - t^{2} - t \tag{3}$$

This turns out to have degree 28. Thus for $k \geq 0$ we will have $0 = u\,P(T)\,T^n\,v = \sum_{i=0}^{28} p_i\, a(i+n)$ where

p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 28, which would hold true for any u and v. It seems that with our particular u and v we have a recurrence of order only 19, corresponding to a factor Q of P.

```
> Q:= t^19 - (11*t^18-11*t^17-195*t^16 +361*t^15 +1387*t^14 -2443*
  t^13 -5105*t^12 +6602*t^11 +9836*t^10
  -7666*t^9 -9322*t^8 +3553*t^7 +4167*t^6 -507*t^5 -741*t^4 +25*t^3
  +51*t^2 -t -1);
```

$$Q := t^{19} - 11\,t^{18} + 11\,t^{17} + 195\,t^{16} - 361\,t^{15} - 1387\,t^{14} + 2443\,t^{13} + 5105\,t^{12} - 6602\,t^{11} - 9836\,t^{10} + 7666\,t^9 + 9322\,t^8 - 3553\,t^7 - 4167\,t^6 + 507\,t^5 + 741\,t^4 - 25\,t^3 - 51\,t^2 + t + 1 \quad (4)$$

The complementary factor R has degree 9.

```
> R:= normal(P/Q);
```

$$R := \left(t^8 + 6\,t^7 + 9\,t^6 - 6\,t^5 - 18\,t^4 + 9\,t^2 - 1\right) t \quad (5)$$

Now we want to show that $b(n) = u\,Q(T)\,T^n\,v = 0$ for all n. This will certainly satisfy the order-9 recurrence

$$\sum_{i=0}^{9} r_i\,b(i+n) = \sum_{i=0}^{9} r_i u\,Q(T)\,T^{n+i} v = u\,Q(T)\,R(T)\,T^n\,v = u\,P(T)\,T^n\,v = 0$$

where r_i are the coefficients of $R(t)$. To show all $b(n) = 0$ it suffices to show $b(0) = \ldots = b(8) = 0$.

```
> uQ:= u . unapply(Q,t)(T):
  seq(uQ . TV[i],i=0..8);
```

$$0, 0, 0, 0, 0, 0, 0, 0, 0 \quad (6)$$